Supporting Information

New *NOAH* Modules for Structure Elucidation at Natural Isotopic Abundance

Ēriks Kupče[1] *and Tim D. W. Claridge*[2]

[1]Bruker BioSpin, Coventry, UK,

[2]University of Oxford, Oxford, UK

Table of Contents

1. Pulse schemes of the HMQC, HSQC, ASAP-COSY and 2BOB modules represented schematically in the main text.

The NOAH modules represented schematically in the main text are shown here in detail in Fig. S1.



Fig. S1. The pulse schemes of the NOAH modules used in the main text – (a) ASAP-COSY, (b) 2BOB, (c) HMQC and (d) HSQC; filled rectangles denote $\pi/2$ pulses and hollow rectangles denote π pulses; all pulses are applied with phase x unless indicated otherwise; ϕ_1 = *x, -x;* ϕ_2 = *x, x, -x, -x;* ϕ_3 = *4x, 4(-x); rec = x, -x, -x, x;* except in (a) and (b) *rec = x, -x;* and in (b) ϕ_1 = *x, -x, -x, x; for X =* ^{13}C the adiabatic (CA-WURST-20) pulses A_1 and A_2 are 1.58 and 0.5 ms long covering 340 *ppm* bandwidth and sweeping from low to high field were generated on Bruker systems using the *WaveMaker* software; delays: constant time delay, T = 20 - 30 *ms*, $\varepsilon = \pi_H/2$, $\tau = (^1J_{min} + {}^1J_{max})^{-1}$, $\tau' = \tau_1 + \tau_2$, $\tau_1 = [4(^1J_{min}+0.146\Delta J)]^{-1}$, $\tau_2 = [4(^1J_{max} - 0.146\Delta J)]^{-1}$, $\Delta J = {}^1J_{max} - {}^1J_{min}$, δ_1 = G1 + recovery time, $\delta_2 = 0$, $\Delta_1 = 0.25/{}^1J_{av}(XH) + \tau_p/2$, $\Delta_2 = 0.25/{}^1J_{av}(XH) - \tau_p/2$, where τ_p is the duration of the *J*-compensated adiabatic WURST pulse (A_1); the dotted rectangle represents $\pi(^1H)$ pulse that is required for optional multiplicity editing with delay δ_2 set to $\delta_2 = 0.5/{}^1J(XH)$, gradients (G/cm, ms): *g0* = (16.5, 1.0), the coherence selection gradients *G1* = (40, 1.0), *G2 = 2G1*γX/γH*, except in (a) *G2 = G1* and in (b) G2 = G1/4, G3 = 3G2; the polarity of the gradient pulses *G1* and the receiver phase were inverted for all even increments; for *X =* ^{15}N the adiabatic pulses are optional and can be replaced with rectangular pulses.

2. Sensitivity Considerations

The sensitivity of the NOAH BSX-type (*X* = COSY, TOCSY, ROESY, NOESY and similar) supersequences has been analysed in ref. [13] (see the main text). Here we provide an example of sensitivity analysis for the NOAH-2 BO supersequence. The sensitivity of the first module, *B* (HMBC) in this example is unaffected by the subsequent module(s) and is assumed to have the same S/N

ratio as that of the equivalent individual pulse sequence. The intensity variations in the spectra generated by the second module, O (2BOB) as compared to the stand-alone pulse sequence are shown in Fig. 2S.



Fig. S2. (a) – (c) NOAH-2 (BO) spectra of Quinine, 17 mg in 600 µL of CDCl3 recorded on a 700 MHz Avance III HD spectrometer, 2 scans per increment, raw data size was 2k x 1k (2k x 512 per module); experiment time was 34 min 42 sec, the total experiment time of individual experiments was 61 min 47 sec, corresponding to the time saving factor, F_t of 1.78; (d) and (e) traces taken along the dashed purple line and comparing signal intensities in the standalone experiments (blue) with the NOAH-2 (BO) spectra (red).

As expected, there is a slight drop of 7 % or less in the amplitudes in both the CT HMQC and the 2D CT HMQC-COSY spectra produced by the 2BOB module due to the imperfections in the magnetization preservation scheme of the preceding ZZ-HMBC module. The sensitivity enhancement of the 2BOB module per unit time, ε_t can be calculated as follows:

$$\varepsilon_t = R_A\sqrt{\rho_t} = 0.93\sqrt{1.78} = 1.24$$

where R_A is the signal amplitude reduction factor and ρ_t is the time saving factor. When ε_t = 1, there is no sensitivity improvement, but the time saving may still be significant. In this NOAH-2 (BO) example the time saving factor is 1.78 and the sensitivity improvement per unit time for the HMBC module is 33 % and for the 2BOB module >24 %. A more extensive analysis of the sensitivity improvements achieved by NOAH supersequences will be published elsewhere.

3. Supplementary Spectra

Supplementary spectra of the compounds used in this work are presented in this section.



Fig. S3. NOAH-4 (BMSC) spectra of ajmalicine in DMSO-d_6 recorded on a 700 MHz Avance III HD spectrometer; a) ^{13}C HMBC, b) ^{15}N HMQC, c) ^{13}C HSQC and d) COSY. For further details see caption to Fig. 2 in the main text.



Fig. S4. NOAH-2 (BO) spectra of gibberellic acid in DMSO-d_6; a) ^{13}C HMBC and b) 2BOB spectra showing overlay of CT HMQC and 2D CT HMQC-COSY spectra; magnitude mode spectra are shown in black, positive peaks in blue and negative (CH_2) peaks in red. For further details see caption to Fig. 3 in the main text.



Fig. S5. Comparison of NOAH-2 SC^2 H-C COSY spectrum (blue) and the conventional ^{13}C HMBC spectrum (black) of pamoic acid in DMSO-d_6 recorded on a 700 MHz Avance III HD spectrometer; the H-C COSY spectrum has been rotated to match the F1 and F2 axis and slightly shifted to the right. Note the $^1J_{CH}$ splittings for protonated carbon sites along the ^{13}C dimension.

4. Pulse Programs Used to Acquire the NOAH Spectra

a. NOAH-4 BMSC pulse program

```
;NOAH-4 BMSC: 13C HMBC + 15N HMQC + 13C me-HSQC + COSY
;Topspin 3 version 2018/04/09
;WaveMaker supported version
;with adiabatic refocussing and optional mult. editing in HSQC
;
;Ref: E. Kupce and T. D. W. Claridge, J. Magn. Reson., 2019,
submitted
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"p4=p3*2"
"p22=p21*2"
"d2=0.5s/cnst2"     ;JCOMP
"d3=0.25s/cnst2"
"d21=0.25s/cnst4"
"d6=0.5s/cnst13"
"d0=3u"
"d10=3u"
"d20=3u"
"in0=inf1/2"
"in10=inf1*cnst10"  ; cnst10 is a scaling factor swC/swH for COSY
"in20=in0*cnst20"   ; cnst20 is a scaling factor swC/swN for 15N
HMQC
"l0=td1/8"
"l9=(d9/(p45*20))"

"DELTA=d2-p16-10u"
"DELTA1=p2+d0*2"
"DELTA2=d6-p16-10u"
"DELTA3=p16+d16+p2/2+4u+d0-p3*2/3.14159"
"DELTA4=d16+p2/2+d0-4u+p21*2/3.14159"
"DELTA5=d21-p16-de-8u+p1*2/3.14159"
"DELTA6=d3-p31/2"
"DELTA7=d3+p31/2"

"DELTA8=p16+d16+p1+4u-p3/3.14159"
"DELTA9=d3-p14/2"
"DELTA10=d3+p14/2"
"DELTA12=DELTA10-p16-p3-de-p2/3.14-8u"

"acqt0=0"
```

```
baseopt_echo

1 ze
  30m pl12:f2
2 30m
3 4m do:f2
4 50u UNBLKGRAD
  4u pl1:f1
  p16:gp0
  4u
  (p1 ph1)
  4u pl3:f3
  p16:gp0*1.37
  50u BLKGRAD
  d1 st0
  p1 ph1         ;ZZ-HMBC
  DELTA6
  (p31:sp18 ph4):f2
  (center (p2 ph1):f1 (p22 ph1):f3)
  DELTA7
  p1 ph1
  DELTA6 UNBLKGRAD
  (p31:sp18 ph4):f2
  (center (p2 ph1):f1 (p22 ph1):f3)
  DELTA7 pl2:f2 ; -HC-Hz
  p1 ph1        ; +HC(z) -Hx
  10u
  p16:gp0
  DELTA          ; J-filter
  (p3 ph1):f2
  10u
  p16:gp0*-1
  DELTA2
  (p3 ph5):f2
  DELTA3
  (p14:sp3 ph4):f2
  4u
  p16:gp1*EA*0.77
  d16
  d0
  (p2 ph1)
  d0
  4u
  p16:gp1*EA*0.77
  d16
  (p14:sp3 ph4):f2
  DELTA3 pl2:f2
  (p3 ph3):f2
  (p2 ph1)
  4u
  p16:gp2*0.77
  4u
  d16 ; BLKGRAD
  goscnp ph30       ;acquire HMBC

  4u pl3:f3
```

```
  4u pl2:f2
  (p21 ph1):f3
  (p3 ph1):f2
  50u
  p16:gp0*1.77
  2m st

5 p1 ph8  ; 15N HMQC starts here
  d21
  (center (p2 ph1):f1 (p22 ph1):f3)
  d21 UNBLKGRAD
  (p1 ph8):f1
  (p21 ph3):f3
  4u
  p16:gp1*-1*EA
  DELTA4
  (p22 ph3):f3
  d20
  p16:gp1*EA
  d16
  (p2 ph4)
  p16:gp1*EA
  d16
  d20
  (p22 ph5):f3
  4u
  p16:gp1*-1*EA
  DELTA4
  (p21 ph5):f3
  d21
  (center (p2 ph1):f1 (p22 ph1):f3)
  4u
  p16:gp5
  DELTA5 4u pl16:f3
  4u
  goscnp ph30 cpd3:f3  ; acquire 15N HMQC
  4u do:f3

  4u pl3:f3
  4u pl2:f2
  (p21 ph1):f3
  (p3 ph1):f2
  50u
  p16:gp0*1.77
  5m st

  (p1 ph1)                        ;HSQC
  DELTA9
  (p14:sp3 ph1):f2
  (p2 ph1):f1
  DELTA10 pl2:f2
  (p1 ph2):f1
  (p3 ph3):f2
  DELTA8
  (p14:sp3 ph1):f2
  4u
```

```
  p16:gp1*EA
  d16
  d0
  (p2 ph1)
  d0
  4u
  p16:gp1*EA
  d16
  (p14:sp3 ph1):f2
                      ;optional HSQC multiplicity editing
#ifdef EDIT
    DELTA8
    (p31:sp18 ph6):f2
    (p2 ph1):f1
    d2
    (p31:sp18 ph6):f2
    (p2 ph1):f1
    d2 pl2:f2
#else
    DELTA8 pl2:f2
#endif

  (p3 ph5):f2
  (p1 ph1):f1
  DELTA9
  (p14:sp3 ph1):f2
  (p2 ph1):f1
  p16:gp2
  DELTA12 ; BLKGRAD
  4u pl2:f2
  (p3 ph1):f2
  4u pl12:f2
  goscnp ph30 cpd2:f2   ;acquire HSQC
  50u do:f2

if "d9 > 1m"   ;ASAP-COSY
{
  50u ; UNBLKGRAD
  p16:gp0*1.77
  d16
  50u ; BLKGRAD

9 (p45:sp45 ph10):f1
  (p45:sp45 ph12):f1
  (p45:sp45 ph11):f1
  (p45:sp45 ph12):f1
  (p45:sp45 ph10):f1
  (p45:sp45 ph10):f1
  (p45:sp45 ph12):f1
  (p45:sp45 ph11):f1
  (p45:sp45 ph12):f1
  (p45:sp45 ph10):f1
  (p45:sp45 ph13):f1
  (p45:sp45 ph15):f1
  (p45:sp45 ph14):f1
  (p45:sp45 ph15):f1
```

```
  (p45:sp45 ph13):f1
  (p45:sp45 ph13):f1
  (p45:sp45 ph15):f1
  (p45:sp45 ph14):f1
  (p45:sp45 ph15):f1
  (p45:sp45 ph13):f1
  lo to 9 times l9
  4u pl1:f1      ;end mixing
}

  50u ; UNBLKGRAD
  p16:gp0
  5m st

  (p1 ph7)
  4u
  p16:gp3*EA
  d16
  d10            ;COSY t1-evolution
  (p1 ph1)
  4u
  p16:gp3
  d16
  go=2 ph31      ;acquire H-H COSY
  30m wr #0 if #0 zd igrad EA

  lo to 3 times 2
  1m id0
  1m id10
  1m id20
  1m ip3*2
  1m ip30*2
  lo to 4 times l0

50u BLKGRAD
exit


ph1=0
ph2=1
ph3=0 2
ph4=0 0 0 0 2 2 2 2
ph5=0 0 2 2
ph6=0
ph7=0 2
ph8=2
ph10=0
ph11=60
ph12=150
ph13=180
ph14=240
ph15=330
ph30=0 2 2 0
ph31=0 2
```

```
;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p14: f2 channel - 180 degree shaped pulse for inversion
;p45: f1 channel -  180 degree CAWURST-2 pulse
;sp3: f2 channel - shaped pulse 180 degree
;sp45: f1 channel - CAWURST-2 pulse (180 degree)
;p16: homospoil/gradient pulse                          [1 msec]
;d0 : incremented delay (2D)                            [3 usec]
;d10 : incremented delay (2D)                           [3 usec]
;d20 : incremented delay (2D)                           [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d9: ASAP mixing duration [0 (no mixing) or 40-60 ms]
;d11: delay for disk I/O                                [5 msec]
;d16: delay for homospoil/gradient recovery
;cnst2: = J(XH)
;cnst13: = nJ(XH) long range
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;ns:  2 * n
;ds: 16
;NBL=3: number of NOAH modules
;FnMODE: echo-antiecho

; ~~~~~~~ use 'wvm -a' command to create these shapes
~~~~~~~~~~~~~~~~~~~
;sp3:wvm:wu180C13: cawurst-20(60 kHz, 0.5 ms; L2H)
;sp18:wvm:wu180Jcomp: wurst-20(340 ppm; L2H, Jcomp, C2=0.385,
C3=116.6)
;cpd2:wvm:wudec: cawurst_d-20(220 ppm, 1.4 ms; L2H)
;sp45:wvm:wuASAP: cawurst-2(30 ppm, 1.0 ms; Q=3)

;use gradient ratio:  gp0 : gp1 : gp2  : gp3  : gp4 : gp5
;                            17.13 : 80  : 40.2 : 40.1 : 33  : 32.4


;for z-only gradients:
;gpz0: 17.13%
;gpz1: 80%
;gpz2: 40.2%
;gpz3: 40.1%
;gpz4: 33%
;gpz5: 32.4%

;use gradient files:
;gpnam0: SMSQ10.100
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100
```

```
;EDIT: for C-13 multiplicity editing start experiment with
;             option -DEDIT (eda: ZGOPTNS)

;Processing
;use splitx_au program to split and process the data
;define user processing au programs (USERP1...P4) for automatic
processing of all data sets
```

b. NOAH-2 BO pulse program

```
;NOAH-2 (BO), HMBC + 2BOB with multiplicity editing
;
;Ref: E. Kupce and T. D. W. Claridge, J. Magn. Reson., 2019,
submitted
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"cnst30=sfo1/(sfo2*4)"
"cnst31=2*sfo2/sfo1"

define list<gradient> EA1 = {cnst30 -cnst30}
define list<gradient> gR = {cnst31}

"p2=p1*2"
"p4=p3*2"
"d2=0.5s/cnst2"  ;JCOMP
"d3=0.25s/cnst2"
"d6=0.5s/cnst13"
"d0=3u"
"d10=3u"

"l0=td1/8"
"l9=0"

"TAU1=0.5s/(cnst6 + 0.146*(cnst7-cnst6))"
"TAU2=0.5s/(cnst7 - 0.146*(cnst7-cnst6))"

"TAU=d16+p16"
"DELTA=d0*2+p2"
"DELTA1=d3-p14/2"
"DELTA2=d3+p14/2"
"DELTA3=TAU+p2/2+4u+d0-p3*2/PI"
"DELTA4=(TAU1+TAU2)/2-TAU"
"DELTA5=DELTA4-p14/2"
"DELTA6=(TAU1-TAU2)/2-p14/2"
"DELTA7=TAU2-TAU"
"DELTA8=DELTA4+p14/2-8u"
"DELTA11=1s/(2*cnst6)-TAU"
"DELTA12=1s/(2*cnst7)-TAU"
"DELTA14=d6-p16-10u"

"TAU3=d2-4u"
"TAU4=d21/2-d2-p31-p3-p2/2"
"TAU5=d2+p2/2"
```

```
"TAU6=d2-p16"
"TAU7=d21/2-d2*2-p31-p2/2"
"TAU8=TAU6-d0*2-p2/2+p4*2/PI"

"d22=d21/2-d2*2-p31-p2/2"
"d29=TAU7+d0*2+p2-p4*2/PI-16u"

"in0=inf1/2"
"in10=inf1/2"
"in22=in0"
"in29=in0"

;"td1=tdmax(td1,d29,in29)"  ; not working

"acqt0=0"
baseopt_echo

1 ze
  30m pl12:f2
2 32m
3 5m
4 5m do:f2
5 50u UNBLKGRAD
  4u pl1:f1
  p16:gp0
  4u
  (p1 ph1)
  4u
  p16:gp0*1.37
  50u BLKGRAD
  d1 st0
  p1 ph1
  DELTA1
  (p14:sp3 ph4):f2   ; Jcomp L2H
  (p2 ph1):f1
  DELTA2
  p1 ph1
  DELTA1 UNBLKGRAD
  (p14:sp3 ph4):f2   ; Jcomp
  (p2 ph1):f1
  DELTA2 pl2:f2 ; -HC-Hz
  p1 ph1        ; +HC(z) -Hx
  DELTA11   ; J-filter
  p16:gp3
  d16 pl2:f2
  (p3 ph10):f2
  DELTA12
  p16:gp4
  d16
  (p3 ph10):f2
  4u
  p16:gp5
  DELTA14
  (p3 ph5):f2
  DELTA3
  (p14:sp3 ph4):f2   ; short 180
```

```
4u
p16:gp1*EA*0.77
d16
d10
(p2 ph1)
d10
4u
p16:gp1*EA*0.77
d16
(p14:sp3 ph4):f2    ; short 180
DELTA3 pl2:f2
(p3 ph10):f2
(p2 ph1)
4u
p16:gp1*gR*0.77
4u
d16 BLKGRAD
goscnp ph30       ; acquire C-13 HMBC

4u pl2:f2
(p3 ph1):f2
50u UNBLKGRAD
4u pl12:f2
p16:gp0*1.77
2m st
20u cpd2:f2

(p1 ph1)
d22
TAU3 do:f2
4u pl2:f2
(p3 ph3):f2
TAU5 pl0:f2
(p31:sp18 ph1):f2   ; Jcomp L2H;

d0
(p2 ph1)
d0

p16:gp1*EA1
TAU8
4u
(p31:sp18 ph1):f2
4u pl2:f2
(p3 ph5):f2

p16:gp1*EA1*-1
TAU6 pl12:f2
d29 cpd2:f2
4u do:f2
4u pl2:f2
(p1 ph2) (p3 ph1):f2 ;

if "l9 %2 == 1"
{
  p16:gp1*0.5
```

```
    d16
    DELTA5 pl0:f2
    (p14:sp3 ph1):f2  ; short ad180  p14:sp3
  }
  else
  {
    DELTA6
    (p14:sp3 ph1):f2  ; short ad180
    p16:gp1*0.5
    d16
    DELTA7
  }
  (p2 ph1)
  p16:gp1*0.75
  d16
  DELTA8 pl12:f2
  4u cpd2:f2
  4u BLKGRAD
  go=2 ph31
  1m do:f2
  30m wr #0 if #0 zd
  1m iu9
  1m igrad EA
  lo to 3 times 2
  1m igrad EA1
  1m id10
  1m ip10*2
  1m ip30*2
  1m ru9
  lo to 4 times 2
  1m id0
  1m dd22
  1m dd29
  1m ip3*2
  1m ip31*2
  lo to 5 times l0

exit


ph1=0
ph2=1
ph3=0 2
ph4=0 0 0 0 2 2 2 2
ph5=0 0 2 2
ph6=2
ph10=0 2
ph30=0 2 2 0
ph31=0 2 2 0

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p16: homospoil/gradient pulse                          [1 msec]
```

```
;p24: f2 channel - 180 degree shaped pulse for refocussing
;     = 2msec for Crp60comp.4
;d0 : incremented delay (2D)                              [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d16: delay for homospoil/gradient recovery
;cnst13: = J(XH) long range
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;ns: 2 * n
;ds: 16
;td1: number of experiments
;FnMODE: echo-antiecho

; ~~~~~~~ use 'wvm -a' command to create these shapes
~~~~~~~~~~~~~~~~~~~
;sp18:wvm:ad180Cref:  cawurst-20(300 ppm; Jcomp, L2H)
;sp3:wvm:ad180Cj:     cawurst(60 kHz, 0.5 ms)
;cpd2:wvm :wudec: cawurst_d-20(220 ppm, 1.4 ms)

;for z-only gradients:
;gpz0:  33%
;gpz1:  80%
;gpz2:  40.1%
;gpz3:  15%
;gpz4: -10%
;gpz5: -5%

;use gradient files:
;gpnam0: SMSQ10.100
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100

;Processing
;use: splitx - to separate JMBC and 2BOB followed by
;split ipap - to separate the time-shared HMQC and HMQC-COSY data of
the 2BOB experiment
```

1.3. NOAH-2 SC^2 pulse program

```
;NOAH-2 (SC2) - me-HSQC + PANSY-COSY
;avance-version (16/07/25)
;Ref: E. Kupce and T. D. W. Claridge, J. Magn. Reson., 2019,
submitted
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>
#include <De.incl>

"p2=p1*2"
"p4=p3*2"
"d2=0.5s/cnst2"
"d3=0.25s/cnst2"
"d0=3u"
"d10=3u"
"in0=inf1/2"
"in10=inf1*cnst10"  ; cnst10 is a scaling factor swC/swH for COSY
"l3=td1/4"

"DELTA=p16+d16+4u+p1-p4/3.14159+2u"
"DELTA1=d3-p14/2"
"DELTA2=d3+p14/2"
"DELTA3=d2+p3+p2"
"DELTA4=DELTA+p3"
"DELTA5=p2"
"DELTA6=DELTA2-p16-p3-de+p1*2/3.14-8u"

"acqt0=0"
baseopt_echo


1 ze1
  ze2
  4u pl1:f1 pl2:f2
2 7m
3 2m
4 50u UNBLKGRAD
  p16:gp0
  (p1 ph2):f1
  (p3 ph2):f2
  p16:gp3*-1
  4u BLKGRAD
  d1 st0

  (p1 ph1)
```

```
  DELTA1
  (p14:sp3 ph6):f2
  (p2 ph1):f1
  DELTA2 pl2:f2 UNBLKGRAD
  (p1 ph2):f1
  (p3 ph3):f2
  DELTA
  (p14:sp3 ph6):f2
  4u
  p16:gp1*EA
  d16
  d0
  (p2 ph4)
  d0
  4u
  p16:gp1*EA
  d16
               ;optional multiplicity editing
#ifdef EDIT
  DELTA3
  (p14:sp3 ph6):f2
  DELTA4 pl2:f2
  (p2 ph2):f1
  d2
  (p3 ph5):f2
  DELTA5
#else
  (p14:sp3 ph6):f2
  DELTA pl2:f2
  (p3 ph5):f2
#endif   ; end EDIT
  (p1 ph1):f1
  DELTA1
  (p14:sp3 ph6):f2
  (p2 ph1):f1
  p16:gp2
  DELTA6 pl2:f2
  4u BLKGRAD
  (p3 ph1):f2
  4u pl12:f2
  goscnp ph29 cpd2:f2  ; HSQC
  4u do:f2
  50u UNBLKGRAD
  p16:gp3
  2m st
  4u pl2:f2
  (p1 ph1):f1
  d2
  (p3 ph6):f2
  10u
  p16:gp3*EA
  d16
  d10
  (p1 ph3):f1
  p16:gp4
  d16
```

```
  (p3 ph3):f2
  p16:gp5
  d16
  4u BLKGRAD
  ACQ_START1(ph28,ph30)
  ACQ_START2(ph28,ph31)
  0.1u DWELL_GEN1 1u DWELL_GEN2
  (aq1) (aq2)
  eoscnp2
  rcyc=2

  3m wr1 #0 if1 #0 zd1
  3m wr2 #1 if2 #1 zd2
  1m igrad EA
  lo to 3 times 2
  1m id0
  1m id10
  lo to 4 times l3
exit

ph1=0
ph2=1
ph3=0 2
ph4=0 0 0 0 2 2 2 2
ph5=0 0 2 2
ph6=0
ph7=0 2
ph28=0
ph29=0 2 2 0
ph30=0 2
ph31=0 2

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;p1 : f1 channel -  90 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p16: homospoil/gradient pulse                          [1 msec]
;d0 : incremented delay (F1 in 3D)                      [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d16: delay for homospoil/gradient recovery
;l3 : loop for 2D experiment = td1
;inf1: 1/SW = 2 * DW
;in0: 1/SW = 2 * DW
;ns: 4 * n
;ds: 32
;td1: number of experiments
;FnMODE: EA in F1
; ~~~~~~~ use 'wvm -a' command to create these shapes
~~~~~~~~~~~~~~~~~~
;sp3:wvm:wu180C13: cawurst-20(80 kHz, 0.5 ms)
;cpd2:wvm:wudec: cawurst_d-20(220 ppm, 1.4 ms)


;for z-only gradients:
;gpz3/gpz1 = 1.063  (gammaH/gammaF) or 0.94 (gammaF/gammaH)
;gpz2+gpz3 = gp1
```

```
;gpz1: 40%
;gpz2: -2.52%  for 1H-obs  or  2.4 19F-obs
;gpz3: 42.52%  for 1H-obs  or 37.6 19F-obs
;gpz4: 7%
;gpz5: 11%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100
```

1.4. NOAH-4 BSCC^2 pulse program

```
;noah4_BSCC2 - 13C HMBC + 13C me-HSQC + H-H COSY + PP/PH PANSY-COSY
;
;WaveMaker supported version with adiabatic refocussing
;with optional multiplicity editing in HSQC
;
;Ref: E. Kupce and T. D. W. Claridge, J. Magn. Reson., 2019,
submitted
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>
#include <De.incl>

"cnst29= sfo2/sfo1"
"cnst28= 1.0-cnst29"

define list<gradient> EA1= {cnst29}
define list<gradient> EA2= {cnst28}

"p2=p1*2"
"d2=1s/(cnst2*2)"    ;JCOMP
"d4=1s/(cnst2*4)"
"d6=1s/(cnst13*2)"

"d0=3u"
"d10=3u"
"d20=3u"
```

```
"in0=inf1/2"
"in10=inf1*cnst10"  ; cnst10 is a scaling factor swC/swH for COSY
"in20=inf1*cnst20"  ; cnst10 is a scaling factor swC/swH for COSY

"l3=td1/8"
"l9=(d9/(p45*20))"

"cnst28=cnst28"   ; for display purposes only
"cnst29=cnst29"   ; for display purposes only

"DELTA=p16+d16+d0+p2/2-p21*2/PI+4u"
"DELTA1=d4-p14/2"
"DELTA2=d4+p14/2"
"DELTA3=d2+p21+p2/2"
"DELTA4=DELTA+p21-p2/2"
"DELTA5=p2"
"DELTA6=DELTA2-p16-p21-de+p1*2/PI-8u"
"DELTA7=d4-p31/2"
"DELTA8=d4+p31/2"
"DELTA11=1s/(2*cnst6)-p16-d16"
"DELTA12=1s/(2*cnst7)-p16-d16"
"DELTA13=d6-p16-d16-4u"

"acqt0=0"
baseopt_echo

1 ze1
  ze2
2 61m
3 6m do:f3
4 50u UNBLKGRAD
  4u pl1:f1 pl3:f3
  p16:gp0
  4u
  (p1 ph2):f1
  (p21 ph2):f3
  4u
  p16:gp0*1.37
  4u BLKGRAD
  4u pl9:f1
  d1 cw:f1 ph29
  4u do:f1
  20u pl1:f1
  5m st0
;  ZZ-HMBC

  (p1 ph1):f1
  DELTA7
  (p31:sp18 ph6):f3
  (p2 ph1):f1
  DELTA8
  (p1 ph1):f1
  DELTA7 UNBLKGRAD
  (p31:sp18 ph6):f3
  (p2 ph1):f1
  DELTA8 pl3:f3
```

```
  p1 ph1
  DELTA11 pl3:f3
  p16:gp3
  d16
  (p21 ph3):f3
  DELTA12
  p16:gp4
  d16
  (p21 ph3):f3
  4u
  p16:gp5
  d16
  DELTA13
  d6

  (p21 ph5):f3
  DELTA
  (p14:sp3 ph6):f3
  4u
  p16:gp1*EA*0.77
  d16
  d0
  (p2 ph4)
  d0
  4u
  p16:gp1*EA*0.77
  d16
  (p14:sp3 ph6):f3
  DELTA pl3:f3
  (p21 ph3):f3
  (p2 ph2)
  4u
  p16:gp2*0.77
  4u
  d16 BLKGRAD
  goscnp ph29

  4u pl3:f3
  (p21 ph1):f3
  50u UNBLKGRAD
  p16:gp0*1.77
  d16
  2m st

; HSQC

  (p1 ph1)
  DELTA1
  (p14:sp3 ph6):f3
  (p2 ph1):f1
  DELTA2 pl3:f3
  (p1 ph2):f1
  (p21 ph3):f3
  DELTA
```

```
#ifdef EDIT
  (p31:sp18 ph1):f3
#else
  (p14:sp3 ph6):f3
#endif

  4u
  p16:gp1*EA
  d16
  d0
  (p2 ph4)
  d0
  4u
  p16:gp1*EA
  d16

#ifdef EDIT
    DELTA3
    (p31:sp18 ph6):f3
    DELTA4
    (p2 ph2):f1
    d2 pl3:f3
    (p21 ph5):f3
    DELTA5
#else
    (p14:sp3 ph6):f3
    DELTA pl3:f3
    (p21 ph5):f3
#endif

  (p1 ph1):f1
  DELTA1
  (p14:sp3 ph6):f3
  (p2 ph1):f1
  p16:gp2
  4u pl3:f3
  DELTA6 BLKGRAD
  (p21 ph1):f3
  4u pl16:f3
  goscnp ph29 cpd3:f3  ; acquire HSQC
  50u do:f3

if "d9 > 1m"   ;ASAP-COSY
{
  50u  UNBLKGRAD
  p16:gp0*1.77
  d16
  50u  BLKGRAD

9 (p45:sp45 ph10):f1
  (p45:sp45 ph12):f1
  (p45:sp45 ph11):f1
  (p45:sp45 ph12):f1
  (p45:sp45 ph10):f1
  (p45:sp45 ph10):f1
  (p45:sp45 ph12):f1
```

```
  (p45:sp45 ph11):f1
  (p45:sp45 ph12):f1
  (p45:sp45 ph10):f1
  (p45:sp45 ph13):f1
  (p45:sp45 ph15):f1
  (p45:sp45 ph14):f1
  (p45:sp45 ph15):f1
  (p45:sp45 ph13):f1
  (p45:sp45 ph13):f1
  (p45:sp45 ph15):f1
  (p45:sp45 ph14):f1
  (p45:sp45 ph15):f1
  (p45:sp45 ph13):f1
  lo to 9 times l9
  4u pl1:f1      ;end mixing
}

  50u UNBLKGRAD
  p16:gp0
  2m st

  (p1 ph3)
  4u
  p16:gp3*EA
  d16
  d10            ;COSY t1-evolution
  (p1 ph1)
  4u
  p16:gp3
  d16 BLKGRAD
  goscnp ph28    ;acquire H-H COSY

  50u UNBLKGRAD
  p16:gp0
  2m st

  (p3 ph3):f2
  50u UNBLKGRAD
  p16:gp6*EA
  d16
  d20
  (p3 ph5):f2
  4u
  p16:gp6*EA2
  d16
  (p1 ph5):f1
  4u
  p16:gp6*EA1
  d16
  4u BLKGRAD     ;acquire P/H PANSY-COSY
  ACQ_START1(ph28,ph30)
  ACQ_START2(ph28,ph31)
  0.1u DWELL_GEN1 1u DWELL_GEN2
  (aq1) (aq2)
  eoscnp2
  rcyc=2
```

```
  30m wr1 #0 if1 #0 zd1
  30m wr2 #1 if2 #1 zd2

  1m igrad EA

  lo to 3 times 2

  1m id0
  1m id10
  1m id20
  1m ip3*2
  1m ip4*2
  1m ip28*2
  1m ip29*2

  lo to 4 times l3

  50u BLKGRAD
exit


ph1=0
ph2=1
ph3=0 2
ph4=0 0 0 0 2 2 2 2
ph5=0 0 2 2
ph6=0
ph10=0
ph11=60
ph12=150
ph13=180
ph14=240
ph15=330
ph28=0 2
ph29=0 2 2 0
ph30=0
ph31=0

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;sp3: f2 channel - adiabatic pulse 180 degree for inversion
;sp18: f2 channel - adiabatic pulse 180 degree for inversion
;sp45: f1 channel - CAWURST-2 pulse (180 degree)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p14: f2 channel - 180 degree shaped pulse for inversion
;p31: f2 channel - 180 degree adiabatic pulse for inversion
;p45: f1 channel -  180 degree CAWURST-2 pulse
;p16: homospoil/gradient pulse                          [1 msec]
;d0 : incremented delay (2D)                            [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)XH
;d4 : 1/(4J)XH
```

```
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d9: ASAP mixing duration [0 (no mixing) or 40-60 ms]
;d16: delay for homospoil/gradient recovery
;aq: acquisition time
;cnst2: = 1J(XH)
;cnst13: = nJ(XH) long range
;l3 : loop for 2D experiment = td1/4
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;nbl: 3 number of NOAH modules
;ns:  1 * n
;ds: 16
;td1: number of experiments - multiple of NBL
;FnMODE: echo-antiecho
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz0: 17.13%
;gpz1:  80%
;gpz2:  40.2% for C-13, 16.2% for N-15
;gpz3:  15%
;gpz4: -10%
;gpz5:  5%

;use gradient files:
;gpnam0: SMSQ10.100
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100

                                        ;preprocessor-flags-start
;EDIT: for C-13 multiplicity editing start experiment with
;         option -DEDIT (eda: ZGOPTNS)
                                              ;preprocessor-flags-end

;WaveMaker shapes
; ~~~~~~~ use 'wvm -a' command to create these shapes ~~~~~~~~~~~~

;sp3:wvm:wu180C13: cawurst-20(220 ppm, 0.5 ms; L2H)
;sp18:wvm:wu180Jcomp: wurst-20(200 ppm; Jcomp, L2H)
;cpd3:wvm:wudec: cawurst_d-20(220 ppm, 1.4 ms; L2H)
;sp45:wvm:wuASAP: cawurst-2(30 ppm, 1.0 ms; Q=3)
```

5. **CMCse Structure Elucidation Results.**

Details

Chemical formula: $C_{21}H_{24}N_2O_3$

Mass [Da]: 352.43
Solvent: DMSO
Description:



Descriptors

InChI: InChI=1/C21H24N2O3/c1-12-16-10-23-8-7-14-13-5-3-4-6-18(13)22-20(14)19(23)9-15(16)17(11-26-12)21(24)25-2/h3-6,11-12,15-16,19,22H,7-10H2,1-2H3

InChIKey: GRTOGORTSDXSFK-UHFFFAOYNA-N

SMILES: O=C(OC)C1=COC(C)C2CN3CCC=4C=5C=CC=CC5NC4C3CC12

Project: C:\Users\eriks.kupce\Desktop\MRdata\ajmalicine_cmc\ajmalicine_cmc

Report file: C:\Users\eriks.kupce\Desktop\MRdata\ajmalicine_cmc\ajmalicine_cmc\cmc_report.pdf



^{1}H table of assignments

Atom	Shift [ppm]	Multiplicity	Bound to	Correlation table
C18'	1.02		(C18)	H20
C21	1.12		(C21)	H19
C17	1.92		(C17)	H18
C14'	2.2		(C14)	H17
C19	2.37		(C19)	H16
C15'	2.57		(C15)	H15
C20'	2.61		(C20)	H14
C20	2.78		(C20)	H13
C14	2.95		(C14)	H12
C15	3.02		(C15)	H11
C18	3.22		(C18)	H10
C13	3.36		(C13)	H9
C16	3.68		(C16)	H8
C12	4.48		(C12)	H7
C7	6.94		(C7)	H6
C6	7.01		(C6)	H5
C9	7.3		(C9)	H4
C8	7.35		(C8)	H3
C2	7.47		(C2)	H2
N22	10.84		(N22)	H1

^{13}C table of assignments

Atoms assigned to fragments are shown in italic.

Atom	Shift [ppm]	# H's	Correlation table
	15.2	3	C21
	22.1	2	C20
	30.62	1	C19
	33.29	2	C18
	41.23	1	C17
	51.21	3	C16
	53.0	2	C15
	56.49	2	C14
	60.18	1	C13
	73.6	1	C12
	106.8	0	C11
	107.58	0	C10
	111.6	1	C9
	117.84	1	C8
	118.75	1	C7
	120.77	1	C6
	127.13	0	C5
	135.84	0	C4
	136.58	0	C3
	154.11	1	C2
	166.89	0	C1

^{1}H table of assignments

Atom	Shift [ppm]	Multiplicity	Bound to	Correlation table
C18'	1.02		(C18)	H20
C21	1.12		(C21)	H19
C17	1.92		(C17)	H18
C14'	2.2		(C14)	H17
C19	2.37		(C19)	H16
C15'	2.57		(C15)	H15
C20'	2.61		(C20)	H14
C20	2.78		(C20)	H13
C14	2.95		(C14)	H12
C15	3.02		(C15)	H11
C18	3.22		(C18)	H10
C13	3.36		(C13)	H9
C16	3.68		(C16)	H8
C12	4.48		(C12)	H7
C7	6.94		(C7)	H6
C6	7.01		(C6)	H5
C9	7.3		(C9)	H4
C8	7.35		(C8)	H3
C2	7.47		(C2)	H2
N22	10.84		(N22)	H1



^{13}C table of assignments

Atoms assigned to fragments are shown in italic.

Atom	Shift [ppm]	# H's	Correlation table
	15.2	3	C21
	22.1	2	C20
	30.62	1	C19
	33.29	2	C18
	41.23	1	C17
	51.21	3	C16
	53.0	2	C15
	56.49	2	C14
	60.18	1	C13
	73.6	1	C12
	106.8	0	C11
	107.58	0	C10
	111.6	1	C9
	117.84	1	C8
	118.75	1	C7
	120.77	1	C6
	127.13	0	C5
	135.84	0	C4
	136.58	0	C3
	154.11	1	C2
	166.89	0	C1



Explained Correlations

HMBC COSY



Incorrect Correlations

HMBC



Chemical Shift Correlation





Details

Chemical formula: $C_{19}H_{22}O_6$

Mass [Da]: 346.38
Solvent: DMSO
Description:



Descriptors

InChI: InChI=1/C19H22O6/c1-9-7-17-8-18(9,24)5-3-10(17)19-6-4-11(20)16(2,15(23)25-19)13(19)12(17)14(21)22/h4,6,10-13,20,24H,1,3,5,7-8H2,2H3,(H,21,22)/f/h21H

InChIKey: IXORZMNAPKEEDV-PKSOQXRJNA-N

SMILES: O=C(O)C1C2C3(OC(=O)C2(C)C(O)C=C3)C4CCC5(O)C(=C)CC14C5

Project: C:\Users\eriks.kupce\Desktop\MRdata\giberellic_BOx\strucelu

Report file: C:\Users\eriks.kupce\Desktop\MRdata\giberellic_BOx\strucelu\cmc_giberellic.pdf



^{1}H table of assignments

Atom	Shift [ppm]	Multiplicity	Bound to	Correlation table
C19	1.07		(C19)	H17
C17'	1.61		(C17)	H16
C18	1.63		(C18)	H15
C15'	1.65		(C15)	H14
C15	1.73		(C15)	H13
C13	1.85		(C13)	H12
C17	1.9		(C17)	H11
C16'	2.14		(C16)	H10
C16	2.19		(C16)	H9
C12	2.49		(C12)	H8
C11	3.07		(C11)	H7
C9	3.87		(C9)	H6
C6'	4.85		(C6)	H5
C6	5.12		(C6)	H4
H3	5.57			H3
C4	5.79		(C4)	H2
C5	6.32		(C5)	H1

^{13}C table of assignments

Atoms assigned to fragments are shown in italic.

Atom	Shift [ppm]	# H's	Correlation table
	14.97	3	C19
	17.02	2	C18
	39.25	2	C17
	43.2	2	C16
	44.75	2	C15
	49.87	0	C14
	50.91	1	C13
	51.3	1	C12
	52.57	1	C11
	53.55	0	C10
	68.91	1	C9
	77.09	0	C8
	91.02	0	C7
	106.72	2	C6
	131.95	1	C5
	133.74	1	C4
	158.15	0	C3
	173.61	0	C2
	179.19	0	C1

^{1}H table of assignments

Atom	Shift [ppm]	Multiplicity	Bound to	Correlation table
C19	1.07		(C19)	H17
C17'	1.61		(C17)	H16
C18	1.63		(C18)	H15
C15'	1.65		(C15)	H14
C15	1.73		(C15)	H13
C13	1.85		(C13)	H12
C17	1.9		(C17)	H11
C16'	2.14		(C16)	H10
C16	2.19		(C16)	H9
C12	2.49		(C12)	H8
C11	3.07		(C11)	H7
C9	3.87		(C9)	H6
C6'	4.85		(C6)	H5
C6	5.12		(C6)	H4
H3	5.57			H3
C4	5.79		(C4)	H2
C5	6.32		(C5)	H1



^{13}C table of assignments

Atoms assigned to fragments are shown in italic.

Atom	Shift [ppm]	# H's	Correlation table
	14.97	3	C19
	17.02	2	C18
	39.25	2	C17
	43.2	2	C16
	44.75	2	C15
	49.87	0	C14
	50.91	1	C13
	51.3	1	C12
	52.57	1	C11
	53.55	0	C10
	68.91	1	C9
	77.09	0	C8
	91.02	0	C7
	106.72	2	C6
	131.95	1	C5
	133.74	1	C4
	158.15	0	C3
	173.61	0	C2
	179.19	0	C1



Explained Correlations

HMBC H2BC



Incorrect Correlations

HMBC



Chemical Shift Correlation